Behavior Research Methods & Instrumentation
1981, Vol. 13 (2), 227-234

SESSION XI
GENERAL APPLICATIONS I

James H. Johnson, *Presider*

A microcomputer-based cardiotachometer with video display

HARDY J. POTTINGER, CARROLL W. HUGHES, PAM SCHROEDER, and ANN BAREFIELD
University of Missouri, Rolla, Missouri 65401

and

JAMES C. CRAIGMILE
University of Missouri, Columbia, Missouri 65201

A KIM-1 single board microcomputer implementation of a cardiotachometer that provides a low-cost versatile heart-rate data acquisition system is described. Unique features include telemetry, a composite video output signal of a video frame displaying elapsed time and heart-rate, and a TV camera signal of the behavior. Data is videotape- and/or audiotape-recorded for further analysis with a larger computer.

There is a need for more research on anxiety using physiological measures (Hodges, 1976). However, according to Sieber, O'Neil, and Tobias (1977, p. 34), "it is not feasible to locate physiological recording equipment in school settings, to have electrodes attached to students while they attempt to do academic work." Obviously, an instrument that does not require electrodes with leads running to a physiograph would be more suitable to naturalistic settings and allow research of a broader range of behavior. The instrument to be described here uses a KIM-1 microprocessor to combine telemetric monitoring of heart rate with a video camera signal. A unique feature of this cardiotachometer is the capability of providing a composite video output signal of a frame displaying elapsed time in minutes and seconds, together with a subject's measured R-R interval. This can be mixed with a television camera signal to provide a composite view of a subject's behavior and his or her heart rate. This apparatus should open new avenues to researchers who wish to use heart rate to monitor the physiological response of subjects in various settings, such as schools and offices, as well as a laboratory environment.

SYSTEM OVERVIEW

Our cardiotachometer is similar in principle to others described recently by Dejong (1980) and Klosterhalfen (1980). The unique feature of the system described here is the combination of telemetry to acquire sensor data and superimposing heart-rate data with a video image of the subject. All of the components, with the exception of the transducer and interface, are standard off-the-shelf components.

An overall view of the system is shown in Figure 1. The input sensor consists of a simple photoplethysmograph mounted on a finger of the subject. The output of the sensor can be either connected directly to the cardiotachometer via suitable signal conditioning electronics or transmitted via a small FM transmitter to a nearby receiver whose output drives the transducer interface. R-wave pulses are amplified, filtered, and peak detected



Figure 1. Cardiotachometer block diagram.

  0005-7878/81/020227-08$01.05/0

by the transducer interface to produce an interrupt request (IRQ) to the 6502 microprocessor.

The number of 10-msec intervals that occur between interrupts are counted by a program residing in nonvolatile read-only memory (1K EPROM). The 10-msec intervals are produced by an on-board crystal-controlled programmable interval timer (PIT). The reciprocal of the R-R intervals are calculated, and a running average of four consecutive rates is maintained. The resulting heart rate in beats per minute is displayed on the integral LED display and output to an alphanumeric frame buffer (MVM-1024). Elapsed time in minutes and seconds is also displayed in the frame buffer for convenience, and the display is updated every 1 sec.

The video output from the frame buffer is mixed with video from a small television camera, and the resulting composite image can either be displayed on a local display or be tape-recorded for off-line analysis.

MICROCOMPUTER-BASED CARDIOTACHOMETER

Microcomputer

The primary component of our system is a KIM-1 single-board computer. The KIM-1 is a familiar component in many instrumentation systems used in psychology and will not be described further here. In order to expand the basic KIM, we used a Riverside Electronics KEM expansion motherboard, an MVM-1024 alphanumeric display module and a KIMSI power supply. Regulators to supply +5, –5, and +12 V dc had to be added to the unregulated power supply. The KEM is an S-100 bus-compatible adaptor for the KIM-1 that also contains room for four 2708 EPROMs. These provide an extra 4,096 bytes of program storage. The cardiotach program and video driver routines are stored in a single 1,024-byte EPROM. This storage is nonvolatile and need not be reloaded after power is removed.

A flowchart of the cardiotachometer program is shown in Figure 2. A complete listing is provided in Appendix A. As seen in Figure 2a, the main timer loop increments a count of the number of 10-msec intervals elapsed since the last interrupt, restarts a 10-msec timer, and displays the current rate on the LED display. The KIM-1 LED display is controlled by software and must be refreshed frequently to provide a flicker-free display. This takes about 2 msec during the 10-msec countdown interval. After 100 10-msec intervals have elapsed, the CRT display is updated with the current rate and elapsed time in minutes and seconds.

Each sensor pulse results in an interrupt that causes control to be passed to IRQ. This routine, shown in Figure 2b, takes the count of the number of 10-msec intervals since the last interrupt, converts it to beats per minute, and computes a running average of the four most recent values. This tends to smooth out short-term variations in R-R intervals and "short beats" due to noise-generated interrupts.



Figure 2. (a) Flowchart of main tachometer program. (b) Interrupt service routine and heart-rate calculation.

Transducer Interface

The transducer interface is shown in Figure 3. This is similar to a circuit used by Dejong (1980) and seems to work reasonably well. A band-pass filter was added to reduce noise from the telemetry system. The threshold detector approach to R-wave detection works fairly well and is simple to implement but must be "tuned" to a particular subject, since there is considerable variation in sensor signal level from subject to subject. A better R-wave detection system is currently being investigated.

VIDEO DISPLAY

The primary component of the video display consists of a Riverside Electronics MVM-1024 microprocessor video display driver. The MVM is capable of displaying 16 rows of 64 alphanumeric characters on a standard television monitor, and contains its own 1,024-character display buffer. We use the lower left-hand corner for display purposes, with the rest of the area available for future display of additional parameters. Subroutines used to drive the MVM-1024 are listed in Appendix B.

The MVM-1024 was slightly modified by the addition of two connectors for horizontal and vertical synchronization signals. The MVM's own sync signals are used to externally synchronize the television camera and thus eliminate the need for a common external synchronization source. A simple resistor network suffices to mix the two video signals to provide a composite data/image video signal for recording and/or display.

INPUT TRANSDUCER

The input transducer shown schematically in Figure 1 consists of a simple photodiode-phototransistor pair (Radio Shack 276-142) mounted inside a small plastic finger-cuff. R1 is approximately 370 ohms to provide a diode current of about 20 mA. Load Resistor R2 is 200 kohms and is not at all critical. Supply voltage (V) is 9 V taken from a small 9-V battery. The output is a pulse of about .2-V pulse-to-pulse (p-p) amplitude superimposed on a 4- to 5-V dc bias.

The finger-cuff and battery are mounted inside a child's glove, together with a small FM transmitter. This apparatus provides a rather unobtrusive instrumentation package. Dummy packages can be made up and passed off as "space commando gloves," so subjects do not know who is being monitored.

As an alternative, various commercial types of finger-cuffs or photoplethysmographs are available (e.g., Lafayette Instrument Company). Some commercial telemetry systems can use a variety of transducers and/or disposable electrode attachments (e.g., Midgard Electronics).

TELEMETRY SYSTEM

An existing Bio-Sentry Telemetry Model 4200 FM instrumentation receiver and Model 201 transmitter are used to telemeter sensor output to the transducer interface. The transmitter operated on IRIG Channel 7. There is nothing at all special about this system, which was used simply because it was available. A smaller transmitter package and one tailored specifically for this application would have been desirable. The current package measures 6.00 x 1.65 x 3.81 cm, excluding the battery and sensor. The Bio-Sentry Telemetry Model 2200 voltage-controlled oscillator and amplifier were originally designed to monitor ECG and had to be preceded with a resistive voltage divider to reduce the sensor output signal to a level compatible with the transmitter. Many alternative telemetry packages are available.

FUTURE PLANS

We plan to use the system in a normal classroom environment to collect data in order to study factors influencing classroom anxiety in children. The system will allow an investigator to conveniently and systematically analyze physiological data (heart rate) and behavior in relationship to events preceding and following significant arousals.

We intend to extend this single-subject pilot model to a multiple-subject system. The use of a microprocessor will allow us to conveniently modify the system with minimum impact on hardware design. Other physiological measures such as GSR and T-wave amplitude can be readily accommodated. We are also currently developing such a system for animals; in the system, heart-rate data are transmitted via an implantable transmitter (e.g., Konigsburg Instruments). The combination of a video image of a subject and a physiological parameter in an unobtrusive manner promises to open up new vistas for the behavioral investigator.

REFERENCES

DeJong, M. L. A digital cardiotachometer implemented with the AIM 65. *Compute II*, August/September 1980, pp. 32-35.

Hodges, L. The psychophysiology of anxiety. In M. Zuckerman & C. D. Spielberger (Eds.), *Emotions and anxiety: New concepts, methods, and applications.* Hillsdale, N.J: Erlbaum, 1976.

Klosterhalfen, W. A computer-controlled cardiotachometer. *Behavior Research Methods & Instrumentation*, 1980, 1, 58-62.

Sieber, J. E., O'Neil, H. F., Jr., Tobias, S. *Anxiety, learning and instruction.* New York: Wiley, 1977.

Appendix A
Cardiotachometer Program Listing

```
CTACH                    KIM-1 CROSS ASSEMBLER

                         ; CTACH
                         ; CARDIOTACHOMETER PROGRAM FOR THE KIM-1 MICROCOMPUTE
                         , 9-1-80
                         ; PAM SCHROETER, UMR EE DEPT
```

```
                        ;
                                .LOC $1000
                        TIMER   EQU $1706
                        TIMOT   EQU     $1707
                        PAD     EQU     $1700
                        PADD    EQU     $1701
                        CLOCK   EQU     $0A
                        SEC     EQU     $0B
                        MIN     EQU     $0C
                        POINTR  EQU     $81
                        COUNTR  EQU     $00
                        CNT     EQU     $03
                        CNTHI   EQU     $02
                        CNTLO   EQU     $01
                        CNTMI   EQU     $03
                        SCANDS  EQU     $1F1F
                        TEMP    EQU     $09
                        STATUS  EQU     $F1
                        ERASE   EQU     $124C
                        BUFFER  EQU     $04
                        PNTL    EQU     $05
                        IN      EQU     $06
                        INH     EQU     $F9
                        POINTL  EQU     $FA
                        POINTH  EQU     $FB
                        VIDSP   EQU     $1200     ; VIDEO DISPLAY DRIVER
                        ;
                        ; INITIALIZATION
                        ;
1000 78                 START   SEI               ; DISABLE INTERUPTS
1001 A9 62                      LDA #$62
1003 8D FE 17                   STA $17FE         ; SET UP INTERUPT VECTOR
1006 A9 10                      LDA #$10
1008 8D FF 17                   STA $17FF
100B A9 00                      LDA #$00
100D 85 F1                      STA STATUS
100F 85 81                      STA POINTR
1011 85 0B                      STA SEC
1013 85 0C                      STA MIN
1015 A2 01                      LDX #$01
1017 8D 00 17                   STA PAD           ; SET UP DATA DIRECTION REG
101A 8D 01 17                   STA PADD
101D A9 64                      LDA #$64
101F 85 0A                      STA CLOCK
1021 20 4C 12                   JSR ERASE         ; ERASE THE SCREEN
1024 CE 00 17                   DEC PAD
1027 A2 FF                      LDX #$FF
1029 86 00                      STX COUNTR
102B 58                         CLI               ; ENABLE INTERUPTS
                        ;
                        ; LOOP TO MEASURE THE TIME BETWEEN EACH HEARTBEAT AND
                        ; MEASURE THE REAL TIME ELAPSED.
                        ; UPDATES PULSE AND TIME EACH SECOND.
                        ;
102C A9 9C              LOOP    LDA #$9C          ; START TIMER FOR 10 MSEC
102E 8D 06 17                   STA TIMER
1031 E6 00                      INC COUNTR
1033 C6 0A                      DEC CLOCK
1035 D0 1D                      BNE NOMIN         ; WAIT 1 SECOND
1037 A9 64                      LDA #$64
1039 85 0A                      STA CLOCK         ; RESET CLOCK
103B 20 00 12                   JSR VIDSP         ; DISPLAY PULSE AND REAL TIME
103E F8                         SED
103F 18                         CLC
1040 A5 0B                      LDA SEC
1042 65 01                      ADC 01            ; INCR SECONDS
1044 C9 60                      CMP #$60
1046 D0 09                      BNE CONT          ; BRANCH IF .LT. 60 SEC
1048 A5 0C                      LDA MIN
104A 18                         CLC
104B 69 01                      ADC #$01          ; INCR MIN
104D 85 0C                      STA MIN
104F A9 00                      LDA #$00
1051 85 0B              CONT    STA SEC           ; SAVE SECONDS
1053 D8                         CLD
1054 20 1F 1F           NOMIN   JSR SCANDS        ; DISPLAY PULSE RATE
1057 20 1F 1F                   JSR SCANDS        ; DO IT AGAIN
105A AD 07 17           CHEK    LDA TIMOT         ; CHECK TIMER, IF NOT FINISHE
105D 10 FB                      BPL CHEK          ; BRANCH TO CHECK AGAIN
```

```
105F 4C 2C 10            JMP LOOP         ; START TIMER AGAIN
                 ;
                 ; IRQ
                 ; EACH PULSE GENERATES AN INTERUPT
1062 EE 00 17    IRQ     INC PAD          ; PAD = 1
1065 A5 00               LDA COUNTR       ; IF COUNTR = 0, GO TO AGN
1067 D0 03               BNE IRQ1
1069 4C                  JMP AGN
106C 85 03       IRQ1    STA CNT
106E A9 00               LDA #$00
1070 85 F9               STA INH
1072 85 FA               STA POINTL
1074 85 FB               STA POINTH
                 ;
                 ; DIVIDE
                 ; FIND RECIPROCAL OF INTER-PULSE INTERVAL
                 ;
1076 A2 0A               LDX #$0A
1078 A9 DC               LDA #$DC
107A 85 02               STA CNTHI
107C A9 05               LDA #$05
107E 85 00               STA 0 CNTLO
1080 38                  SEC
1081 E5 03               SBC CNT
1083 30 15       BRANCH  BMI NEG
1085 CA          POS     DEX
1086 D0 03               BNE DIV1
1088 4C                  JMP ENDIV
108B 85 01       DIV1    STA CNTLO
108D 38                  SEC
108E 26 02               ROL CNTHI
1090 26 01               ROL CNTLO
1092 A5 01               LDA CNTLO
1094 38                  SEC
1095 E5 03               SBC CNTMI
1097 4C 83 10            JMP BRANCH
109A CA          NEG     DEX
109B D0 03               BNE DIV2
109D 4C                  JMP ENDIV
10A0 85 01       DIV2    STA CNTLO
10A2 18                  CLC
10A3 26 02               ROL     CNTHI
10A5 26 01               ROL CNTLO
10A7 A5 01               LDA CNTLO
10A9 18                  CLC
10AA 65 03               ADC CNTMI
10AC 4C 83 10            JMP BRANCH
10AF A6 04       ENDIV   LDX BUFFER
                 ;
                 ; BUFSTO
                 ; STORE THE RESULT IN A BUFFER AT $0004
                 ;
10B1 C5 81       BUFSTO  CMP POINTR
10B3 D0 04               BNE BST1
10B5 A5 00               LDA COUNTR
10B7 85 81               STA POINTR
10B9 A5 02       BST1    LDA CNTHI
10BB A6 81               LDX POINTR
10BD 95 04               STA BUFFER,X
10BF E6 81               INC POINTR
                 ;
                 ; ADD
                 ; ADD FOUR BYTES ($0004 - $0007),
                 ; FORMING THE AVERAGE OF FOUR BEATS
                 ;
10C1 18          ADD     CLC
10C2 A9 00               LDA #$00
10C4 A2 04               LDX #$04
10C6 75 04       ADLOOP  ADC BUFFER,X
10C8 CA                  DEX
10C9 D0 FB               BNE ADLOOP
10CB 20 DE 10    ADL1    JSR CNVT
10CE A5 05               LDA PNTL
10D0 85 FA               STA POINTL
10D2 A5 06               LDA IN
10D4 85 F9               STA INH
                 ;
                 ; AGN
                 ; RE-INIT THE COUNTER
```

```
                        ;
10D6 CE 00 17           AGN     DEC PAD
10D9 A2 FF                      LDX #$FF
10DB 86 00                      STX COUNTR
10DD 40                         RTI
                        ;
                        ; CNVT
                        ; BINARY TO DECIMAL CONVERT
                        ; TAKES 8-BIT NUMBER IN THE ACC AND CONVERTS IT
                        ; TO A DECIMAL NUMBER WITH LOW ORDER STORED IN 'IN'
                        ; HIGH ORDER STORED IN PNTL.
                        ;
10DE A2 64              CNVT    LDX #$64
10E0 86 09                      STX TEMP
10E2 20 FE 10                   JSR SUBT
10E5 86 05                      STX PNTL
10E7 A2 0A                      LDX #$0A
10E9 86 09                      STX TEMP
10EB 20 FE 10                   JSR SUBT
10EE 86 06                      STX IN
10F0 06 06                      ASL IN
10F2 06 06                      ASL IN
10F4 06 06                      ASL IN
10F6 06 06                      ASL IN
10F8 18                         CLC
10F9 65 06                      ADC IN
10FB 85 06                      STA IN
10FD 60                         RTS
                        ;
10FE A2 00              SUBT    LDX #$00
1100 38                 SUB     SEC
1101 E5 09                      SBC TEMP
1103 90 04                      BCC SUB1
1105 E8                         INX
1106 4C 00 11                   JMP SUB
1109 18                 SUB1    CLC
110A 65 09                      ADC TEMP
110C 60                         RTS
                        ;
                                .END
0  UNDEFINED SYMBOLS **
```

Appendix B
Video Display Driver Subroutine Listings

```
VIDSP                     KIM-1 CROSS ASSEMBLER

                          ; VIDSP
                          ; VIDEO DISPLAY ROUTINE FOR CARDIOTACH
                          ;
                                  .LOC $1200
                          INIT    EQU $135E
                          ERASE   EQU $132F
                          CURHI   EQU $13FA
                          CURLO   EQU $13F9
                          PUTC    EQU $1300
                          SEC     EQU $0B
                          MIN     EQU $0C
                          PLSLO   EQU $0D
                          PLSHI   EQU $0E
                          ;
1200 A9 0E                VIDSP   LDA #$0E
1202 8D FA 13                     STA CURHI      ; SET CURSOR TO DISPLAY AREA
1205 A9 00                        LDA #$00
1207 8D F9 13                     STA CURLO
120A A5 0C                        LDA MIN        ; GET MINUTES VALUE
120C 20 31 12                     JSR DISP2      ; DISPLAY 2 DIGITS
120F A9 20                        LDA #$20       ; GET SPACE
1211 20 00 13                     JSR PUTC       ; DO IT
1214 A5 0B                        LDA SEC        ; GET SECONDS VALUE
1216 20 31 12                     JSR DISP2
                          ;
                          ; DISPLAY PULSE RATE
1219 A9 0F                        LDA #$0F
121B 8D FA 13                     STA CURHI
121E A9 00                        LDA #$00
1220 8D F9 13                     STA CURLO
```

```
                  ;
1223 A5 0E                LDA PLSHI
1225 20 31 12             JSR DISP2
1228 A5 0D                LDA PLSLO
122A 20 31 12             JSR DISP2
122D 20 46 12             JSR UBLK
1230 60                   RTS
                  ;
                  ; DISPLAY A 2-DIGIT DECIMAL NUMBER AT CURSOR LOC
                  ; DECIMAL VALUE IS IN ACC
1231 48           DISP2   PHA
1232 29 F0                AND #$F0
1234 4A                   LSR A
1235 4A                   LSR A
1236 4A                   LSR A
1237 4A                   LSR A
1238 49 30                EOR #$30
123A 20 00 13             JSR PUTC
123D 68                   PLA
123E 29 0F                AND #$0F
1240 49 30                EOR #$30         ; MAKE IT AN ASCII DIGIT
1242 20 00 13             JSR PUTC
1245 60                   RTS
                  ;
                  ; UNBLANK THE SCREEN
1246 A9 10        UBLK    LDA #$10
1248 8D FA 13             STA CURHI
124B 60                   RTS
                  ;
                  ; ERASE THE SCREEN UTILITY ROUTINE
124C 20 5E 13             JSR INIT
124F 20 2F 13             JSR ERASE
1252 00                   BRK
                          .END
0 UNDEFINED SYMBOLS **
                  ; VIDRV
                  ; DRIVER ROUTINE FOR MVM1024 VIDEO DISPLAY
                  ;
                  ; ROUTINES ARE:
                  ;       PUTC - OUTPUT CHAR IN ACC
                  ;       ERASE - CLEAR SCREEN
                  ;       INIT - INITIALIZE VIDEO DISPLAY
                  ;
                  ; DEFINE I/O PORT LOCATIONS
                  ;
                  CURHI   EQU     $13FA
                  CURLO   EQU     $13F9
                  DATA    EQU     $13FB
                  ;
                          .LOC $1300
                  ;
                  ; PUTC
                  ; OUTPUT CHAR IN ACC
                  ; UPDATE CURSOR LOCATION
                  ; INTERPRET CR ($0D) TO BE NEW LINE
1300 C9 0D        PUTC    CMP #$0D         ; CR?
1302 F0 10                BEQ CRFND        ; YES
1304 8D FB 13             STA DATA         ; NO, JUST DISPLAY IT
                  ;
1307 AD F9 13             LDA CURLO
130A 29 3F                AND #$3F         ; CHECK FOR END OF LINE
130C C9 3F                CMP #$3F
130E F0 04                BEQ CRFND        ; END OF LINE, SAME AS CR
1310 EE F9 13             INC CURLO        ; NOPE, JUST INCREMENT COL
1313 60                   RTS              ; RETURN
                  ;
1314 A9 00        CRFND   LDA #$00
1316 8D F9 13             STA CURLO        ; NEW LINE
1319 AD FA 13             LDA CURHI
131C 29 0F                AND #$0F
131E C9 0F                CMP #$0F         ; CHECK FOR END OF SCREEN
1320 D0 06                BNE OK
1322 A9 10                LDA #$10
1324 8D FA 13             STA CURHI
1327 60                   RTS
                  ;
1328 18           OK      CLC
1329 69 11                ADC #$11
132B 8D FA 13             STA CURHI
```

```
132E 60                    RTS
                    ;
                    ; CLEAR SCREEN
132F A9 00          ERASE  LDA #00
1331 8D FA 13              STA CURHI
1334 8D F9 13              STA CURLO
                    ;
1337 A9 20          LP1    LDA #$20
1339 A2 40                 LDX #$40
133B 8D FB 13       LOOP   STA DATA
133E EE F9 13              INC CURLO
1341 CA                    DEX
1342 D0 F7                 BNE LOOP
                    ;
1344 A9 00                 LDA #$00
1346 8D F9 13              STA CURLO
1349 EE FA 13              INC CURHI
134C AD FA 13              LDA CURHI
134F 29 0F                 AND #$0F
1351 D0 E4                 BNE LP1
1353 A9 10                 LDA #$10
1355 8D FA 13              STA CURHI
1358 A9 00                 LDA #$00
135A 8D F9 13              STA CURLO
135D 60                    RTS
                    ;
                    ; INITIALIZE
135E D8             INIT   CLD
135F A2 04                 LDX #$04
1361 A9 00          IN1    LDA #$00
1363 9D FA 13              STA CURHI,X
1366 A9 1C                 LDA #$1C
1368 9D FB 13              STA DATA,X
136B CA                    DEX
136C CA                    DEX
136D 10 F2                 BPL IN1
136F 60                    RTS
                    ;
                    ; DIAGNOSTIC ROUTINE
                    ; DISPLAY ALL CHARACTERS
1370 20 5E 13       TEST   JSR INIT
1373 20 2F 13              JSR ERASE
1376 A9 00                 LDA #$00
1378 85 00                 STA 0           ; USE LOC ZERO
137A A2 FF                 LDX #$FF
137C A5 00          TST1   LDA 0
137E E6 00                 INC 0
1380 20 00 13              JSR PUTC
1383 CA                    DEX
1384 D0 F6                 BNE TST1
1386 00                    BRK             ; RETURN TO MONITOR
                           .END
0  UNDEFINED SYMBOLS **
```